

BenderOrszag Ex12 p299 07-20-16

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[3]:=

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
  StyleDefinitions -> Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Original notebook — *BenderOrszag Ex12 p299 10-03-15*

Purpose

The problem is to determine the leading order asymptotic expansion as $k \rightarrow \infty$ of the integral

$$I(k) = \int_0^{\infty} dz e^{-k(\cosh(z) + t^2/2)} \cos(\pi k z)$$

I Example 12, p 299


The integrand is oscillatory.

In[5]:=

```
Module[{lab, standardArgs},
  F[z_, k_] := Exp[-k (Cosh[z] + z^2 / 2)] Cos[k π z];

  lab = Stl@StringForm["`", TraditionalForm[F[z, k]]];
  Plot[{F[z, 1], F[z, 3], F[z, 5]}, {z, 0, 2}, AxesLabel → {Stl["z"], Stl["F"]},
    PlotLabel → lab, PlotStyle → {Black, Blue, Green},
    PlotLegends → Placed[{"k = 1", "k = 3", "k = 5"}, Right], PlotRange → All]
```

Out[5]=



The standard form I have been using for steepest descent problems is

$$\int dz e^{k\rho(z)} = \int dz e^{k(\phi(z) + i\psi(z))}$$

Rewrite the integral in a form conducive to steepest descent analysis

In[6]:=

```
w1[1] = Exp[-k (Cosh[z] + z^2 / 2)] Cos[k π z] /.
  Cos[k π z] → (Cos[k π z] // TrigToExp) // ExpandAll
```

Out[6]=

$$\frac{1}{2} e^{-i k \pi z - \frac{k z^2}{2} - k \text{Cosh}[z]} + \frac{1}{2} e^{i k \pi z - \frac{k z^2}{2} - k \text{Cosh}[z]}$$

I perform some operations to transform these integrals over these two terms into a single integral that is in the standard form for a steepest descent calculation

In[7]:=

```
w1[2] = Int[#, {z, 0, ∞}] & /@ w1[1]
```

Out[7]=

$$\text{Int}\left[\frac{1}{2} e^{-i k \pi z - \frac{k z^2}{2} - k \text{Cosh}[z]}, \{z, 0, \infty\}\right] + \text{Int}\left[\frac{1}{2} e^{i k \pi z - \frac{k z^2}{2} - k \text{Cosh}[z]}, \{z, 0, \infty\}\right]$$

In[8]:=

```
w1[3] = w1[2] /.
  w1[2][[2]] → (w1[2][[2]] /. z → -z /. Int[a_, {-z, 0, ∞}] → Int[a, {z, -∞, 0}])
```

Out[8]=

$$\text{Int}\left[\frac{1}{2} e^{-i k \pi z - \frac{k z^2}{2} - k \text{Cosh}[z]}, \{z, 0, \infty\}\right] + \text{Int}\left[\frac{1}{2} e^{-i k \pi z - \frac{k z^2}{2} - k \text{Cosh}[z]}, \{z, -\infty, 0\}\right]$$

In[9]:= $w1[4] = w1[3] /. \text{Int}[a_, \{z, 0, \infty\}] + \text{Int}[b_, \{z, -\infty, 0\}] \rightarrow \text{Int}\left[\frac{a+b}{2}, \{z, -\infty, \infty\}\right]$

Out[9]:= $\text{Int}\left[\frac{1}{2} e^{-i k \pi z - \frac{k z^2}{2} - k \text{Cosh}[z]}, \{z, -\infty, \infty\}\right]$

In[10]:= $w1[5] = w1[4] /. \text{Int}[a_. b_, \text{lim}_] /; \text{FreeQ}[a, z] \rightarrow a \text{Int}[b, \text{lim}]$

Out[10]:= $\frac{1}{2} \text{Int}\left[e^{-i k \pi z - \frac{k z^2}{2} - k \text{Cosh}[z]}, \{z, -\infty, \infty\}\right]$

In this case $\rho(z)$ is explicitly given by

In[11]:= $w1[6] = \rho[z] == (w1[5] /. a_ \text{Int}[\text{Exp}[b_], \text{lim}_] \rightarrow \text{Simplify}[b/k])$

Out[11]:= $\rho[z] == -i \pi z - \frac{z^2}{2} - \text{Cosh}[z]$

The equation for the saddle point is

In[12]:= $w1[7] = (D[w1[6], z] /. \rho'[z] \rightarrow 0)$

Out[12]:= $0 == -i \pi - z - \text{Sinh}[z]$

This has no closed form solution


In[13]:= $\text{Solve}[w1[7], z]$

Solve: This system cannot be solved with the methods available to Solve.

Out[13]:= $\text{Solve}[0 == -i \pi - z - \text{Sinh}[z], z]$

But it can be graphically established that a root exists at $z = -i \pi$.

In[14]=



Out[14]=

Now $\rho(z) = \phi(z) + i \psi(z)$, and the curves of constant ψ that satisfy $\psi(z) = \psi(z_{\text{saddlepoint}})$ include the curves of steepest descent

In[15]=

```
w1[8] = w1[7][[2]] /. z -> x + I y // ComplexExpand // ExpandAll
```

Out[15]=

$$-i \pi - x - i y - i \cosh[x] \sin[y] - \cos[y] \sinh[x]$$

In[16]=

```
w1[9] = ψ[x, y] == Simplify[Im[w1[8]], {x ∈ Reals, y ∈ Reals}]
```

Out[16]=

$$\psi[x, y] == -\pi - y - \cosh[x] \sin[y]$$

At the saddle point

In[17]=

```
w1[10] = w1[6] /. z -> -I π
```

Out[17]=

$$\rho[-i \pi] == 1 - \frac{\pi^2}{2}$$

which is real. Thus $\psi(-i\pi) = 0$ and the constant ψ curves passing through the saddle point satisfy the equation

In[18]=


```
w1[11] = w1[9] /.  $\psi[x, y] \rightarrow 0$ 
```

Out[18]=

```
 $0 == -\pi - y - \text{Cosh}[x] \text{Sin}[y]$ 
```

An exact closed form solution is unavailable but RegionPlot can be used to illustrate the constant ψ curves.

```
Module[{X = 2  $\pi$ , Y = 2  $\pi$ , image = 300,  $\mathcal{R}$ , g},
   $\mathcal{R}$  = ImplicitRegion[ $0 == -\pi - y - \text{Cosh}[x] \text{Sin}[y]$ , {x, y}];
  g[1] = RegionPlot[ $\mathcal{R}$ , PlotRange -> {{-X, X}, {-Y, Y}},
    Axes -> True, ImageSize -> image, AxesLabel -> {Stl["x"], Stl["y"]},
    PlotLabel -> Stl[TraditionalForm[- $\pi - y - \text{Cosh}[x] \text{Sin}[y] == 0$ ]],
    FrameTicks -> {{-2  $\pi$ , - $\pi$ , 0,  $\pi$ , 2  $\pi$ }, {-2  $\pi$ , - $\pi$ , 0,  $\pi$ , 2  $\pi$ }}];
  g[2] =
  Module[{k = 2, Z = 0.001, gSurface, lab, F,  $\psi$ Constant},
    F[z_, k_] :=  $\frac{1}{2} e^{-i k \pi z - \frac{k z^2}{2} - k \text{Cosh}[z]}$ ;
    gSurface = ContourPlot[Abs[F[x + I y, k]], {x, -X, X},
      {y, -Y, Y}, ImageSize -> image, MeshFunctions -> {#3 &}, Mesh -> 10,
      PlotRange -> {{-X, X}, {-Y, Y}, {-Z, Z}}, PlotLegends -> Automatic,
      FrameTicks -> {{-2  $\pi$ , - $\pi$ , 0,  $\pi$ , 2  $\pi$ }, {-2  $\pi$ , - $\pi$ , 0,  $\pi$ , 2  $\pi$ }}];
  Grid[{{g[1], g[2]}}]
```



The contour plot shows that the steepest descent curve is the one passing from $\{-2\pi, -2\pi\}$ through the saddle point at $\{0, -\pi\}$ to $\{2\pi, 0\}$.

Examine the region near the saddle point.

In[20]:= `w1[12] = Normal@Series[w1[11], {y, -π, 3}]`

Out[20]=
$$\theta = (\pi + y) (-1 + \text{Cosh}[x]) - \frac{1}{6} (\pi + y)^3 \text{Cosh}[x]$$


Solve for y

In[21]:= `w1[13] = Solve[w1[12], y]`

Out[21]=
$$\left\{ \left\{ y \rightarrow -\pi \right\}, \left\{ y \rightarrow \text{Sech}[x] \left(-\pi \text{Cosh}[x] - 2\sqrt{3} \sqrt{\text{Cosh}[x]} \text{Sinh}\left[\frac{x}{2}\right] \right) \right\}, \right. \\ \left. \left\{ y \rightarrow \text{Sech}[x] \left(-\pi \text{Cosh}[x] + 2\sqrt{3} \sqrt{\text{Cosh}[x]} \text{Sinh}\left[\frac{x}{2}\right] \right) \right\} \right\}$$

Note that it is the third branch that corresponds to the steepest descent curve.

In[22]:= `Plot[w1[13][[3, 1, 2]], {x, -π, π}]`



For $x \ll 1$

In[23]:= `w1[14] = Map[(#[[1, 1]] == Normal@Series[#[[1, 2]], {x, 0, 2}]) &, w1[13]`

Out[23]=
$$\{y = -\pi, y = -\pi - \sqrt{3} x, y = -\pi + \sqrt{3} x\}$$

I compare the steepest descent curve and its $x \ll 1$ approximation

In[24]=


```

Module[{k, X =  $\pi$ , Y =  $2\pi$ , Z = 1.1,  $\delta F = 0.01$ , image = 300, yC, yCapprox, sdCurve1,
  sdCurve2, sdCurveApprox1, sdCurveApprox2, saddlePoint, gSurface, lab, F},
  F[z_, k_] :=  $e^{-i k \pi z - \frac{k z^2}{2} - k \text{Cosh}[z]}$ ;
  yC[x_] :=  $\text{Sech}[x] \left( -\pi \text{Cosh}[x] + 2 \sqrt{3} \sqrt{\text{Cosh}[x]} \text{Sinh}\left[\frac{x}{2}\right] \right)$ ;
  yCapprox[x_] :=  $-\pi + \sqrt{3} x$ ;

  sdCurve1 = {Directive[Yellow, Thick], Arrowheads[{{0.0, 0.05, 0.05, 0.0}}, Arrow@
    Table[{x, yC[x]}, {x, -X, X, 0.1}], {Black, Text[Stl["C"], {-2, yC[-2]}]}};
  sdCurveApprox1 = {Directive[Yellow, Dashed], Arrowheads[{{0.0, 0.05, 0.05, 0.0}},
    Arrow@Table[{x, yCapprox[x]}, {x, -X, X, 0.1}],
    {Black, Text[Stl["Capprox"], {-1.5, yCapprox[-1.5]}]}};
  saddlePoint = {Yellow, PointSize[0.025], Point[{0, - $\pi$ ]}];
  lab = Stl[Stl[TraditionalForm[F[z, k]]]];
  k = 2;
  gSurface =
  ContourPlot[Abs[F[x + I y, k]], {x, -X, X}, {y, -2  $\pi$ , 0}, ImageSize  $\rightarrow$  image,
    MeshFunctions  $\rightarrow$  {#3 &}, Mesh  $\rightarrow$  50, PlotRange  $\rightarrow$  {{-X, X}, {-2  $\pi$ , 0}, {0, Z}},
    Epilog  $\rightarrow$  {saddlePoint, sdCurve1, sdCurveApprox1},
    FrameLabel  $\rightarrow$  {{Stl[Rotate["y", - $\pi/2$ ]], ""}, {Stl["x"], lab}}]]

```

Out[24]=



2 Calculation

The leading order asymptotic approximation is obtained by integrating along the straight line approximation of the steepest descent curve passing through the saddle point $z_{SP} = -i\pi$.

$$I(k) = \int_0^{\infty} dz e^{-k(\cosh(z) + i^2/2)} \cos(\pi k z)$$

$$\simeq \frac{1}{2} \int_{C_{\text{approx}}} dz e^{-i k \pi z - \frac{kz^2}{2} - k \text{Cosh}[z]}$$

Near the saddle point $z = 0$, $\rho(z)$ is approximated by

In[25]:= $w2[1] = \rho[z] \rightarrow \text{Normal@Series}\left[-i \pi z - \frac{z^2}{2} - \text{Cosh}[z], \{z, -i \pi, 5\}\right]$

Out[25]:= $\rho[z] \rightarrow 1 - \frac{\pi^2}{2} + \frac{1}{24} (i \pi + z)^4$

Recall w1[5]

In[26]:= $w1[5]$

Out[26]:= $\frac{1}{2} \text{Int}\left[e^{-i k \pi z - \frac{kz^2}{2} - k \text{Cosh}[z]}, \{z, -\infty, \infty\}\right]$

In[27]:= $w2[2] = \frac{1}{2} \text{Int}\left[e^{k \rho[z]} dz, \{z, -\infty, \infty\}\right] /. w2[1]$

Out[27]:= $\frac{1}{2} \text{Int}\left[dz e^{k \left(1 - \frac{\pi^2}{2} + \frac{1}{24} (i \pi + z)^4\right)}, \{z, -\infty, \infty\}\right]$

In preparation for writing z in polar form write

In[28]:= $w2[3] = w2[2] /. a_ \text{Int}[b_, \{z, -\infty, \infty\}] \rightarrow a \text{Int}[b, \{z, -\infty, \theta\}] + a \text{Int}[b, \{z, \theta, \infty\}]$

Out[28]:= $\frac{1}{2} \text{Int}\left[dz e^{k \left(1 - \frac{\pi^2}{2} + \frac{1}{24} (i \pi + z)^4\right)}, \{z, \theta, \infty\}\right] + \frac{1}{2} \text{Int}\left[dz e^{k \left(1 - \frac{\pi^2}{2} + \frac{1}{24} (i \pi + z)^4\right)}, \{z, -\infty, \theta\}\right]$

In the first integral $z = -i\pi + r e^{i\pi/4}$ and in the second integral $z = -i\pi + r e^{-i\pi/4}$

In[29]:= $w2[4] = (w2[3][[1]] /. \{z, \theta, \infty\} \rightarrow \{r, \theta, \infty\} /. z \rightarrow -i \pi + r \text{Exp}[i \pi / 4] /. dz \rightarrow dr \text{Exp}[i \pi / 4]) +$
 $(w2[3][[2]] /. \{z, -\infty, \theta\} \rightarrow \{r, -\infty, \theta\} /. z \rightarrow -i \pi + r \text{Exp}[-i \pi / 4] /. dz \rightarrow dr \text{Exp}[-i \pi / 4])$

Out[29]:= $\frac{1}{2} \text{Int}\left[dr e^{-\frac{i \pi}{4} + k \left(1 - \frac{\pi^2}{2} - \frac{r^4}{24}\right)}, \{r, -\infty, \theta\}\right] + \frac{1}{2} \text{Int}\left[dr e^{\frac{i \pi}{4} + k \left(1 - \frac{\pi^2}{2} - \frac{r^4}{24}\right)}, \{r, \theta, \infty\}\right]$

In[30]:= $w2[5] = w2[4] /. \text{Int} \rightarrow \text{Integrate} /. dr \rightarrow 1$

Out[30]:= $\text{ConditionalExpression}\left[\frac{\left(-\frac{3}{2}\right)^{1/4} e^{k - \frac{k \pi^2}{2}} \text{Gamma}\left[\frac{1}{4}\right]}{4 k^{1/4}} - \frac{(-1)^{3/4} \left(\frac{3}{2}\right)^{1/4} e^{k - \frac{k \pi^2}{2}} \text{Gamma}\left[\frac{1}{4}\right]}{4 k^{1/4}}, \text{Re}[k] > \theta\right]$

In[31]=

```
w2[6] = Simplify[w2[5], {Re[k] > 0}]
```

Out[31]=

$$\frac{3^{1/4} e^{k-\frac{k\pi^2}{2}} \text{Gamma}\left[\frac{1}{4}\right]}{2 \times 2^{3/4} k^{1/4}}$$

I check this against a numerical evaluation of the original integral.

In[32]=

```
Module[{IAsymptotic, numericalValues, points, lab},
  IAsymptotic[k_] :=  $\frac{3^{1/4} e^{k-\frac{k\pi^2}{2}} \text{Gamma}\left[\frac{1}{4}\right]}{2 \times 2^{3/4} k^{1/4}}$ ;
  numericalValues = Table[{k,
    NIntegrate[Exp[-k (Cosh[z] + z^2/2)] Cos[k π z], {z, 0, 10}], {k, 1, 3, 0.25}];
  points = OpenCircle[#] & /@ numericalValues;
  lab = Stl@StringForm[
    "Comparison of asymptotic expansion and numerical integration\n`",
    TraditionalForm[ $\frac{3^{1/4} e^{k-\frac{k\pi^2}{2}} \text{Gamma}\left[\frac{1}{4}\right]}{2 \times 2^{3/4} k^{1/4}}$ ],
    HoldForm[Integrate[Exp[-k (Cosh[z] + z^2/2)] Cos[k π z], {z, 0, 10}]]];
  Plot[IAsymptotic[k], {k, 1, 5}, Epilog -> points,
  PlotLabel -> lab, AxesLabel -> {Stl["k"], ""}]
```

Comparison of asymptotic expansion and numerical integration

$$\frac{\sqrt[4]{3} e^{k-\frac{k\pi^2}{2}} \Gamma\left(\frac{1}{4}\right)}{2 \times 2^{3/4} \sqrt[4]{k}} \int_0^{10} \exp\left(-k\left(\cosh(z) + \frac{z^2}{2}\right)\right) \cos(k\pi z) dz$$

Out[32]=

